Title	双曲形偏微分方程式の数値解法プログラムの研究開発
Author(s)	坂本, 正雄
Citation	大阪大学大型計算機センターニュース. 1990, 79, p. 53-60
Version Type	VoR
URL	https://hdl.handle.net/11094/65900
rights	
Note	

双曲型偏微分方程式の数値解法プログラムの研究開発

大阪大学工学部精密工学科
坂　本　正　雄

1．概　　要

結晶性固体の塑性変形は主に転位の運動特性により記述される。転位の運動方程式は振動弦モデルによると次のように近似される。

$$M\frac{\partial^2 x}{\partial t^2}=\Gamma\frac{\partial^2 x}{\partial z^2}+b\tau-B\frac{\partial x}{\partial t} \qquad (1)$$

ここで簡単のために転位はほぼ z 軸に平行であるとし、その形状を $x(z)$ とした。Mは転位の単位長さ当りの有効質量、Γは転位の線張力、bはバーガース・ベクトルの長さ、τは有効応力、Bは動摩擦係数、t は時間である。なお結晶の幅は ℓ、つまり $0\leqq z\leqq \ell$ とした。また鏡像力のために転位は結晶表面に垂直であるとの境界条件を課すと

$$\left(\frac{\partial x}{\partial z}\right)_{z=0}=\left(\frac{\partial x}{\partial z}\right)_{z=\ell}=0 \qquad (2)$$

である。

τは外部応力や、溶質原子、他の転位等の存在のために、場所 (x, z) により変動するので、ここでは $b\tau=f(x, z)$ と記す。

今 $x=\alpha bX$、$z=\beta bZ$、$t=\beta^2Bb^2T/\Gamma$ と変数を置き替えると(1)式は

$$A\frac{\partial^2 X}{\partial T^2}+\frac{\partial X}{\partial T}-\frac{\partial^2 X}{\partial Z^2}=F(X, Z) \qquad (3)$$

$$A=\frac{M\Gamma}{\beta^2B^2b^2},\qquad F(X, Z)=\frac{\beta^2 b}{\Gamma\alpha}f(\alpha bX, \beta bZ) \qquad (4)$$

となる。ここで α, β は任意定数、(X, Z) は無次元座標、T は無次元時間である。境界条件は、

$$\left(\frac{\partial X}{\partial Z}\right)_{Z=0}=\left(\frac{\partial X}{\partial Z}\right)_{Z=LZ}=0 \qquad (5)$$

ただし $LZ=\ell/\beta b$ である。すなわち、減衰項を含む、任意力作用下の、境界を固定しない弦（転位）の運動は、(3), (4), (5)式を解けばよい。

本研究開発では $X(Z, T)$ を解としてもつ双曲型偏微分方程式(3), (4)式を(5)式の境界条件と、

$$\left(\frac{\partial X}{\partial T}\right)_{T=0}=V(Z, T=0) \qquad (6)$$

の初期条件の下で解く、FORTRAN77　SX　スーパーコンピューター用のサブルーチンを開発した。ここでV（Z，T）は無次元速度である。なお解は実数解のみを扱っている。(4)式のF（X，Z）は、α，β，bを適当な値にとって、XおよびZのきざみを1とすることにより2次元配列F（J，Z）を作って

$$F(X, Z) = F(J, Z) + \{F(J+1, Z) - F(J, Z)\}(X-J) \quad (7)$$

という直線補間で与えている。ここでJ＝INT（X）であり、Z＝0，1，2，……，LZである。

2．使用法

CALL　PDEHT（F，LX，LZ，A，DT，IT，I，X，XO，XN，V，VMAX，XMIN，XMAX，IER）

である。引数の説明等は表1の通りである。

表1．引　数　の　説　明　等

引数	型	種類・寸法	内　　容
F	単精度実数型	2次元配列	入力。F（0：LX，0：LZ）の任意数値データ配列。保存される。
LX	整　数　型		入力。配列F（0：LX，0：LZ）の寸法宣言子。LX≧1。上限はメモリーサイズにより制限される。保存される。
LZ	整　数　型		入力。配列F（0：LX，0：LZ）の寸法宣言子。LZ≧1。上限はメモリーサイズにより制限される。保存される。
A	倍精度実数型		入力。双曲型偏微分方程式(3)式の係数A。A＞0。保存される。
DT	倍精度実数型		入力。時間のきざみ幅。$0 < DT < \sqrt{A}$であれば解は収束するが、DTが大きいほど解の精度が悪いので注意が必要。保存される。
IT	整　数　型		入力。計算の繰り返し数。経過時間はIT＊DTとなる。IT≧1。保存される。
I	整　数　型		出力。計算打切時の繰り返し数が入る。経過時間はI＊DTとなる。正常終了時はI＝ITである。

引数	型	種類・寸法	内　　　　容
X	倍精度実数型	1次元配列	入出力。X（－1：LZ＋1）の1次元配列。入力データは座標X（J），J＝0，1，2，……，LZを与える。0≦X（J）＜LXでなければならない。計算終了後は出力データが入る。
XO	倍精度実数型	1次元配列	作業用配列。XO（0：LZ)。
XN	倍精度実数型	1次元配列	作業用配列。XN（－1：LZ)。
V	倍精度実数型	1次元配列	入出力。V（0：LZ）の1次元配列。入力データは速度V（J），J＝0，1，2，……，LZを与える。計算終了時は結果が出力される。
VMAX	倍精度実数型		出力。計算途中における速度Vの絶対値の最大を与える。VMAXは1／DTより十分小さいほど解の精度がよい。
XMIN	倍精度実数型		出力。X又はXNで現れた最小値。0≦XMIN＜LXの必要がある。
XMAX	倍精度実数型		出力。X又はXNで現れた最大値。0≦XMAX＜LXの必要がある。
IER	整　数　型		出力。エラーインディケーター

現在センターライブラリーは、SX77で直接使用できるようにはなっていない。そこで本ライブラリ・サブプログラムPDEHTを使用するためには、そのソース・プログラムをユーザーのソース・プロクラムに結合しなければならない。例えば

ACOSのTSSにおいて

```
OLD MAIN;LIBSOURCE/CLIB7/PDEHT, R
RESE
RESAVE MAIN
```

とすればよい。ここでMAINはユーザーの行番付のソース・プログラム、RESAVE後のMAINはMAINにPDEHTを結合したユーザーのソース・プログラムである。

3．エラーインディケーターIERの説明

表1のエラーインディケーターIERの内容は次の通りである。

ＩＥＲ＝０　正常終了

１　Ａ≦０である。処理を中断する。

２　ＬＸ＜１である。処理を中断する。

３　ＬＺ＜１である。処理を中断する。

４　０＜ＤＴ＜√Ａ　でない。処理を中断する。

５　ＩＴ＜１である。処理を中断する。

６　入力、Ｘ（Ｊ），Ｊ＝０，１，２，……，ＬＺの最小値をＸＭＩＮ，最大値をＸＭＡＸとしたとき、ＸＭＩＮ＜０又はＸＭＡＸ≧ＬＸである。処理を中断する。

７　ＸＮ（Ｊ），Ｊ＝０，１，２，……，ＬＺの最小値をＸＭＩＮ、最大値をＸＭＡＸとしたとき、ＸＭＩＮ＜０又はＸＭＡＸ≧ＬＸである。処理を中断する。Ｉには中断時の計算繰返し数が入る。

８　ＶＭＡＸは１／ＤＴより大きい。解の精度が良くないので処理を中断する。Ｉには中断時の計算繰返し数が入る。

４．計算方法

中心差分直接解法により(3)式を解く。(3)式はＤＴを無次元時間きざみ、ｍ回計算後の時間ＴをＴ＝ｍＤＴとすると、Ｚ方向のきざみを１，$X(Z,\ T) = X_{z,m}$ として

$$X_{z,m+1} = \frac{2DT^2}{(DT+2A)}\left\{ X_{z+1,m} - \left[2-\frac{2A}{DT^2}\right] X_{z,m} + X_{z-1,m} + F(X_{z,m}, Z)\right\} + \frac{DT-2A}{DT+2A} X_{z,m-1} \qquad (8)$$

と表せる。境界条件は $X_{-1,m} = X_{1,m}$ ，$X_{LZ+1,m} = X_{LZ-1,m}$ である。解Ｘ（Ｚ，Ｔ）が発振しないための収束条件は０＜ＤＴ＜√Ａである。なお無次元速度Ｖ（Ｚ，Ｔ）は

$$V(Z,\ T) = \frac{1}{2DT}(X_{z,m+1} - X_{z,m-1}) \qquad (9)$$

で与えている。

５．ＳＸのＴＳＳの簡易形による使用例

ソース・プログラム

```
SYSTEM ?FRT7 O MAIN
*LIST
C#SXRUN MAIN:FREE2 OBSERVE
      IMPLICIT REAL*8 (A-H,O-Z)
      PARAMETER(LX=10,LZ=3)
      REAL*4 F(0:LX,0:LZ)
      REAL*8 X(-1:LZ+1),V(0:LZ),XO(0:LZ),XN(-1:LZ)
      DO 10 K=0,LZ
        DO 10 J=0,LX
          F(J,K)=1.D-1
   10 CONTINUE
      DO 20 K=0,LZ
        X(K)=5.D0
        V(K)=1.D-1
   20 CONTINUE
      A=1.0D2
      DT=5.D-1
      IT=2
      WRITE(6,*) (X(J),V(J),J=0,LZ)
      DO 40 K=1,3
        CALL PDEHT(F,LX,LZ,A,DT,IT,I,X,XO,XN,V,VMAX,XMIN,XMAX,IER)
        WRITE(6,30) IER,DT,VMAX,XMIN,XMAX,(X(J),V(J),J=0,LZ)
   30   FORMAT(1H0,I2,4E13.6/(1H ,6E13.6))
        IF(IER.NE.0) STOP
   40 CONTINUE
      STOP
      END
C
      SUBROUTINE PDEHT(F,LX,LZ,A,DT,IT,I,X,XO,XN,V,VMAX,XMIN,XMAX,IER)
      IMPLICIT REAL*8 (A-H,O-Z)
      REAL*4 F(0:LX,0:LZ)
      REAL*8 X(-1:LZ+1),V(0:LZ),XO(0:LZ),XN(-1:LZ)
      IF(A.LE.0.D0) GO TO 100
      IF(LX.LT.1) GO TO 200
      IF(LZ.LT.1) GO TO 300
      DSA=DSQRT(A)
      IF(DT.LE.0.D0.OR.DT.GE.DSA) GO TO 400
      IF(IT.LT.1) GO TO 500
      XMIN=X(0)
      XMAX=X(0)
      DO 1 J=1,LZ
        IF(X(J).LT.XMIN) XMIN=X(J)
        IF(X(J).GT.XMAX) XMAX=X(J)
    1 CONTINUE
      IF(XMIN.LT.0.D0.OR.XMAX.GE.DFLOAT(LX)) GO TO 600
      IER=0
      I=0
      VMAX=DABS(V(0))
      VSCZ=2.D0*DT**2/(DT+2.D0*A)
      VSC4T=(2.D0*A-DT)/(2.D0*A+DT)
      VSA=VSC4T/VSCZ
      VSB=2.D0-(1.D0+VSC4T)/VSCZ
      X(-1)=X(1)
      X(LZ+1)=X(LZ-1)
      DO 10 J=0,LZ
        NI=X(J)
        XD2=X(J)-NI
        FV=(F(NI+1,J)-F(NI,J))*XD2+F(NI,J)
        XO(J)=VSCZ/(1.D0+VSCZ*VSA)*(X(J+1)-VSB*X(J)+X(J-1)+FV)
     &     -2.D0*DT*V(J)/(1.D0+VSCZ*VSA)
        XN(J)=2.D0*DT*V(J)+XO(J)
        IF(XN(J).LT.XMIN) XMIN=XN(J)
        IF(XN(J).GT.XMAX) XMAX=XN(J)
        IF(DABS(V(J)).GT.VMAX) VMAX=DABS(V(J))
   10 CONTINUE
      IF(XMIN.LT.0.D0.OR.XMAX.GE.DFLOAT(LX)) GO TO 700
      IF(VMAX.GT.1.D0/DT) GO TO 800
      DO 20 J=0,LZ
        XO(J)=X(J)
        X(J)=XN(J)
```

```
   20 CONTINUE
      DO 30 I=1,IT
        X(-1)=X(1)
        X(LZ+1)=X(LZ-1)
        DO 40 J=0,LZ
          NI=X(J)
          XD2=X(J)-NI
          FV=(F(NI+1,J)-F(NI,J))*XD2+F(NI,J)
          XN(J)=VSCZ*(X(J+1)-VSB*X(J)+X(J-1)-VSA*XO(J)+FV)
          IF(XN(J).LT.XMIN) XMIN=XN(J)
          IF(XN(J).GT.XMAX) XMAX=XN(J)
          V(J)=5.D-1/DT*(XN(J)-XO(J))
          IF(DABS(V(J)).GT.VMAX) VMAX=DABS(V(J))
          XO(J)=X(J)
          X(J-1)=XN(J-1)
   40   CONTINUE
        X(LZ)=XN(LZ)
        IF(XMIN.LT.0.D0.OR.XMAX.GE.DFLOAT(LX)) GO TO 710
        IF(VMAX.GT.1.D0/DT) GO TO 810
   30 CONTINUE
   60 DO 50 J=0,LZ
        X(J)=XO(J)
   50 CONTINUE
      RETURN
  100 IER=1
      WRITE(6,101) IER,A
  101 FORMAT(1H0,'** IER=',I1,'  A=',D23.16,' **')
      RETURN
  200 IER=2
      WRITE(6,201) IER,LX
  201 FORMAT(1H0,'** IER=',I1,'  LX=',I11,' **')
      RETURN
  300 IER=3
      WRITE(6,301) IER,LZ
  301 FORMAT(1H0,'** IER=',I1,'  LZ=',I11,' **')
      RETURN
  400 IER=4
      WRITE(6,401) IER,DT,DSA
  401 FORMAT(1H0,'** IER=',I1,'  DT=',D23.16,'  DSQRT(A)=',D23.16,' **')
      RETURN
  500 IER=5
      WRITE(6,501) IER,IT
  501 FORMAT(1H0,'** IER=',I1,'  IT=',I11,' **')
      RETURN
  600 IER=6
      WRITE(6,601) IER,XMIN,XMAX,I
  601 FORMAT(1H0,'** IER=',I1,'  XMIN=',D23.16,'  XMAX=',D23.16,'  I=',
     &  I11,' **')
      RETURN
  700 IER=7
      WRITE(6,601) IER,XMIN,XMAX,I
      RETURN
  710 IER=7
      WRITE(6,601) IER,XMIN,XMAX,I
      GO TO 60
  800 IER=8
      WRITE(6,801)IER,VMAX,I
  801 FORMAT(1H0,'** IER=',I1,'  VMAX=',D23.16,'  I=',I11,' **')
      RETURN
  810 IER=8
      WRITE(6,801)IER,VMAX,I
      GO TO 60
      END
```

実行例

```
*SXRUN
5.000000000000000  0.1000000000000000  5.000000000000000  0.1000000000000000
5.000000000000000  0.1000000000000000  5.000000000000000  0.1000000000000000

 0 0.500000E+00 0.100000E+00 0.500000E+01 0.515000E+01
 0.510000E+01 0.100000E+00 0.510000E+01 0.100000E+00 0.510000E+01 0.100000E+00
 0.510000E+01 0.100000E+00

 0 0.500000E+00 0.100000E+00 0.510000E+01 0.525000E+01
 0.520000E+01 0.100000E+00 0.520000E+01 0.100000E+00 0.520000E+01 0.100000E+00
 0.520000E+01 0.100000E+00

 0 0.500000E+00 0.100000E+00 0.520000E+01 0.535000E+01
 0.530000E+01 0.100000E+00 0.530000E+01 0.100000E+00 0.530000E+01 0.100000E+00
 0.530000E+01 0.100000E+00
*
```

謝　辞

本研究開発課題は大阪大学大型計算機センター研究開発計画の一環として行われた。